CR-171641

Final Report

CONTRACT NAS9-16606

# COMPUTER MODELING AND SIMULATION OF DUAL PASSAGE HEAT PIPES DURING STEADY-STATE OPERATION

TO

NASA Johnson Space Center

ON

January 13, 1983

BY

George P. Peterson
Assistant Professor
Texas A&M University



THROUGH THE

Texas Engineering Experiment Station

The Texas A&M University System

(NASA-CR-171641) COMPUTER MODELING OF HEAT PIPE PERFORMANCE Final Report (Texas A&M Univ.) 26 p HC A03/MF A01 CSCL 20D

N83-24806

Unclas
G3/34 11736

| 1. Report No. | 2. Government Accession No. | 3. Recipient's Catalog No. |
|---|---|---|
| 4. Title and Subtitle<br>Computer Modeling of Heat Pipe Performance | | 5. Report Date<br>January 13, 1983<br>6. Performing Organization Code<br>NASA |
| 7. Author(s)<br>George P. Peterson | | 8. Performing Organization Report No. |
| 9. Performing Organization Name and Address<br>Engineering Technology Dept.<br>Texas A&M University<br>College Station, Texas 77843 | | 10. Work Unit No.<br>11. Contract or Grant No<br>NAS 9-16606 |
| 12. Sponsoring Agency Name and Address<br>National Aeronautics and Space Administration<br>Johnson Space Center<br>Houston, Tx. 77058 | | 13. Type of Report and Period Covered<br>Final Report<br>14. Sponsoring Agency Code |

15. Supplementary Notes

Research Advisor: Gary Rankin

16. Abstract

Presented herein is a parametric study of the defining equations which govern the steady state operational characteristics of the Grumman Monogroove Dual Passage Heat Pipe. These defining equations are combined to develop a mathematical model which describes and predicts the operational and performance capabilities of a specific heat pipe,given the necessary physical characteristics and working fluid.

Included is a brief review of the current literature , a discussion of the governing equations,and a description of both the mathematical and computer model. Final results of preliminary test runs of the model are presented and compared with experimental tests performed by Grumman on actual prototypes.



| 17. Key Words (Suggested by Author(s)) | 18. Distribution Statement |
|---|---|
| Heat Pipes<br>Computer Modeling<br>Monogroove<br>Dual Passage | |

| 19. Security Classif. (of this report) | 20. Security Classif. (of this page) | 21. No. of Pages | 22. Price* |
|---|---|---|---|
| unclassified | unclassified | | |

*For sale by the National Technical Information Service, Springfield, Virginia 22161

JSC Form 1424 (Rev Nov 75) NASA — JSC

## INTRODUCTION

Presently, the primary means for rejecting heat from orbiting spacecraft is through a space radiator system composed of a series of fluid loops. These loops circulate fluid through the radiator panels which in turn reject heat to the space environment. Because the current system uses a mechanically pumped coolant circuit to transfer heat throughout the radiating surface, it results in a system whose long mission reliability is low and one vulnerable to complete failure due to penetration by a single meteoroid. Reliability can be increased through the use of redundant plumbing, pumping, and valving hardware, resulting in a large increase in total system weight. Hence,there is a need for significant technical improvements in the development of a long life heat-rejection system which is suitable f - long term,high power missions and can be constructed and deployed on orbit.

One solution to this problem is the development of a large modular radiator system that can be assembled during orbit from a number of standard components. This space-constructable radiator system would fulfill the needs and demands of large long-lived heat rejection systems and would allow systems to be built up to any desired heat load capacity.

The key component of this concept as it is presently conceived is an innovative, high-capacity, dual passage heat pipe designed by Grumman Aerospace Corporation (1). This heat pipe with radiator fins attached,would be "plugged in" to contact heat exchangers providing heat removal from a centralized heat transport loop. This type of system would be insensitive to complete failure due to micrometeoroid puncture, with the puncture of any single heat pipe resulting in only the loss of that module's 2-kilowatt capacity. The damaged segments could be removed by the Orbiter and replaced or repaired as necessary.

The basic design of this improved high-performance dual passage, heat-pipe consists of two large axial channels, one for vapor flow and another for liquid flow (see Figure 1). These two channels are separated by a small "monogroove" slot which creates a high capillary pressure difference and causes liquid to be pumped from the liquid channel to the circumferential grooves in the vapor channel. This configuration permits the axial transport and radial heat trans fer phases to be handled independently resulting in a high axial heat transport capability.

The initial effort in the development of this heat pipe at NASA-JSC has been concentrated on a feasibility demonstration of the dual passage concept. Recently, investigation has been undertaken on the priming capabilities and behavioral characteristics of the liquid-vapor interface configuration during subjection to low-g or zero-g environments, similar to those which would be encountered during the operation of a low orbit Space Operation Center.




Figure 1 Grumman Monogroove Heat Pipe

This particular report formulates a mathematical model and computer program which describe the operational and performance characteristics of the Grumman dual passage heat pipe. This model allows us to predict the steady state performance when given the necessary physical parameters. This model can be used for the following:

- To support the desigining and testing of laboratory test elements and prototypes.
- To define the operating limits of the system.
- To verify and correlate the data of element and prototype tests under 1-G and extrapolate them to 0-G environments.
- To analyze the effect of varying heat source and sink temperatures on the thermal performance of the system.
- To predict and simulate the thermal performance of a heat pipe radiator system operated under orbital environments.

Computer modeling and simulation of heat pipes is a relatively new area with a majority of the work having been accomplished over the last 10-12 years. Early modeling of heat pipes was accomplished by S.W.Chi (2,3) with later contributions made by Thrush et al. (4) and D.K. Anand(5). Frank (6) has developed a generalized heat pipe equation and optimization method for grooved heat pipes which provides a methodology for determining optimum groove dimensions. Finally, Holm and Miller (7) completed a parametric study of the defining equations for heat pipe operation to predict the performance characteristics from one which is dimensionally similar using similarity relations.

The fundamental technique used in a majority of the works cited above was that of solving the pressure equations simultaneously. A similar approach has been adopted here.

## ANALYSIS

Alario et al (1) have presented the two differential pressure balance relationships which govern the performance of a monogroove heat pipe.

$$\Delta P_{\text{wall}} = \Delta P_{\substack{\text{vapor}\\\text{channel}}} + \Delta P_{\substack{\text{liquid}\\\text{channel}}} + \Delta P_{\substack{\text{wall}\\\text{wick}}} + \Delta P_{\substack{\text{head}\\\text{tilt}}} + \Delta P_{\substack{\text{head}\\\text{dia}}} \qquad \text{(A)}$$

$$\Delta P_{\substack{\text{mono}\\\text{groove}\\\text{cap}}} = \Delta P_{\substack{\text{vapor}\\\text{channel}}} + \Delta P_{\substack{\text{liquid}\\\text{channel}}} + \Delta P_{\substack{\text{head}\\\text{tilt}}} \qquad \text{(B)}$$

Equation (A) assures that the wall wick capillary pressure rise is sufficient to overcome the total viscous pressure losses in the vapor channel, liquid channel and circumferential wall grooves, plus the gravity head losses associated with the inside diameter of the vapor channel along with any elevation difference between the evaporator and condenser. In this particular investigation the primary concern lies in the performance during zero-g operation resulting in a simplification of this equation.

Equation (B) examines the pressure change resulting from the monogroove slot to insure that sufficient pressure is developed, to overcome the vapor and liquid viscous losses, plus the gravity loss due to adverse tilt. Simultaneous solution of equations (A) and (B) provides a method for determining the maximum heat transfer, Q. Equations (1) through (8) are as presented by Alario et. al. and describe the individual pressure difference terms as a function of the physical geometry.

$$\Delta P_{\text{wall cap}} = 2\,\sigma \cos\,(\phi + \alpha_w)\,/W_w \tag{1}$$

$$\Delta P_{\text{head dia}} = \rho_L\;D_V \tag{2}$$

$$\Delta P_{\text{head tilt}} = \rho_L\;h \tag{3}$$

$$\Delta P_{\substack{\text{mono}\\ \text{groove}}} = 2\,\sigma \cos\,(\theta + \alpha_g)\,/W_g \tag{4}$$

$$\Delta P_{\substack{\text{vapor}\\ \text{channel}\\ \text{(laminar)}}} = \frac{2\,(fRe)\mu H_V\;\;QL_{EFF}}{g_c\;\rho_V\;\lambda\;A_V\;D_V^{\;2}} \tag{5}$$

$$\Delta P_{\substack{\text{vapor}\\ \text{channel}\\ \text{(turbulent)}}} = \frac{2\,(CRe^m)\;\;Q^2L_{EFF}}{g_c\;\rho v\;\lambda^2 A_V^{\;2}\;D_V} \tag{6}$$

$$\Delta P_{\substack{\text{liquid}\\ \text{channel}}} = \frac{2(fRe)\mu_L\;\;QL_{EFF}}{g_c\;\rho_L\;\lambda\;A_L\;D_L^{\;2}} \tag{7}$$

$$\Delta P_{\substack{\text{wall}\\ \text{wick}}} \quad \frac{32\,\mu_L\;\pi\;D_V\;FQ}{g_c\;\;\rho_L\;\lambda\;\;8} \left[\frac{1}{(nA_W\;\;D_W^2\;NL)_{EVAP}} + \frac{1}{(nA_W D_W^2 NL)_{COND}}\right] \tag{8}$$

ORIGINAL PAGE IS
OF POOR QUALITY

Limitations on the heat pipe can be determined as follows:

"When both of the differential pressure relationships are satisfied, the maximum heat transfer is governed by the wall wick capillary structure and the heat pipe performance is wall-wick limited. However, if the monogroove slot cannot sustain the necessary capillary pressure rise (i.e., large slot gap) then the heat transfer is prematurely limited."

THE PROGRAM

Essentially, the program is designed to solve equations (1) through (8) as a function of Q, substituting those values into equations (A) and (B) and then solving equations (A) and (B).

Various checks are made throughout the program to determine if the heat pipe will prime properly in zero-g, if the flow is laminar or turbulent and if the sonic limit or entrainment limits have been exceeded.

Table 1 shows the necessary input parameters, while Table 2 lists the computer nomenclature used in the program. A strong effort was made to use nomenclature consistent with that used by Grumman in their work in order to avoid confusion. This was accomplished by using a nomenclature listing obtained through Paul Marshal of NASA. In some instances it was necessary to deviate from this list either because the author was unaware of the existance of the Grumman term or a different approach was used requiring additional terms.

Figure 2 illustrates a flow chart of the program and is self explanatory.

Numerous comments are included throughout the program which should help clarify the various steps, and an effort was made to structure the program in an orderly fashion. Appendix A contains a complete listing of the program, while Figure 3 is a copy of the output for a specific trial run.

PROGRAM CAPABILITY

Based upon the input physical parameters such as evaporator length, condensor length, vapor passage diameter, liquid passage diameter etc., this program is capable of determing the maximum heat transfer capacity in watts and

and the transport capacity in watt-meters. In addition,the user can determine if the sonic limit or entrainment limits have been reached, whether the heat pipe is wall wick or monogroove limited, if the heat pipe will prime properly in zero-g and the comparative values of the gravity head, monogroove head and the net capillary rise.

Through the use of this program and a simple incremental loop,*** the effect of each of the input parameters can be determined individually answering such questions as "What happens if the evaporator length is increased?" or "What if the vapor diameter is decreased?" This technique although not a sophisticated optimization technique, can provide valuable information as to the importance of the various physical parameters.

PROGRAM VERIFICATION

Preliminary testing of the program was accomplished by comparing computer predictions with actual experimental results available to the author. Figure 3 is an example of one such trial run. The experimental results of actual prototype tests were very limited, but initial verification tests indicate that the program predicts the heat transfer capacity with reasonable accuracy with all deviation from actual results occurring on the low side. That is to say the program underestimates the actual capacity to a small degree.

In order to determine an actual percentage error in the computer predictions, additional experimental data would need to be obtained from Grumman and comparisons made.

*** This has been done and will be discussed in the oral presentation.

## TABLE 1: INPUT VALUES

| | | |
|---|---|---|
| PROPS (12) | Fluid Properties at OPTEMP | |
| XLAT | Latest Heat of Vaporization | KJ/kg |
| RHOL | Liquid Density | $Kg/m^3$ |
| RHOV | Vapor Density | $kg/m^3$ |
| TCONL | Thermal Conductivity Liquid | $W/m^oC$ |
| XMUL | Liquid Viscosity | Centi Poise |
| XMUV | Vapor Viscosity | Centi Poise |
| PSAT | Saturation Pressure | Bar |
| CP | Specific Heat Constant P | $KJ/kg\ ^oC$ |
| STEN | Surface Tension | N/M |
| TCRIT | Critical Temp | $^oC$ |
| PCRIT | Critical Pressure | Bar |
| RMW | Gas Constant X Molecular Weight | $kj/kg\ ^oK$ |

### GEOMETRY

| | | |
|---|---|---|
| EVAPM | Evaporator Section Length | M |
| TRANSM | Transfer Section Length | M |
| CONDM | Condensor Section Length | M |
| DV | Diameter Vapor Tube | MM |
| DL | Diameter Liquid Tube | MM |
| TW | Wall Thickness | MM |
| YS | Yield Strength | |
| AZ | Liquid Area Fraction | % |
| THETWR | Fluid Wetting Angle | Radians |
| TILT | Tilt Height | MM |
| TW | Wall Thickness | MM |

## TABLE 2: COMPUTER NOMENCLATURE

### GEOMETRY

| | | |
|---|---|---|
| LEFFM | Effective Length | M |
| LVM | Overall Length | M |
| | | |
| IXX | Index | |
| WW | Dummy Variables Web | |
| RW | Root | |
| GD | Groove | |
| XX | | |
| SLPHTR | Taper Angle of Wall Wick | Radians |
| AW | Wetted Area Dummy | |
| WPW | Wetted Penmeter Dummy | |
| DW | Wetted Diameter Dummy | |
| AWE | Wetted Area Evaporator | $M^2$ |
| WPWE | Wetted Perimeter Evaporator | M |
| DWE | Wetted Diameter Evaporator | M |
| AWC | Wetted Area Condenser | $M^2$ |
| WPWC | Wetted Perimeter Condenser | M |
| DWC | Wetted Diameter Condenser | M |
| | | |
| OD | Outside Diameter | MM |
| ODM | Outside Diameter | M |

### HYDRAULIC DIAMETERS

Change from mm to m

| | |
|---|---|
| WPL | Wetted Perimeter Liquid |
| AL | Area Liquid Channel |
| DLH | Diameter Liquid Channel |
| WPV | Wetted Perimeter Vapor Channel |
| AV | Area Vapor Channel |
| DVH | Diameter Vapor Channel |

## VARIABLES

### WALL WICK

| | | |
|---|---|---|
| WWE | Web Width Evaporator | m m |
| RWE | Root Width Evaporator | m m |
| GDE | Groove Depth Evaporator | m m |
| TPIE | Thds/inch Evaporator | thd/in |
| NE | Number of Evaporator Feeds | |
| WWC | Web Width Condenser | m m |
| RWC | Root Width Condenser | m m |
| GDC | Groove Depth Condenser | m m |
| TPIC | Thds/inch Condenser | Thds/in |
| NC | Number of Condenser Feeds | |
| QZ | Form Factor for Heat Transfer | |

### MONOGROOVE

| | | |
|---|---|---|
| WM | Monogroove Width | m m |
| ALPAD | Monogroove Taper Angle | Degrees |
| DELT | Monogroove Stand Off Distance | m m |
| GAMMA | $^{CP}/_{CV}$ | |
| G | Acceleration of Gravity | $m/s^2$ |

## HEAD LOSS COEFFICIENTS

| | | |
|---|---|---|
| RL | $\ell/\rho\ell\lambda$ Dummy Variable | |
| CA | Liquid Channel LC | $\frac{ks}{kj \cdot ms}^2$ |
| CWE | Wall Wick LC Evaporator | " |
| CWC | Wall Wick LC Condenser | " |
| CW | Wall Wick LC Total | " |
| PCW | $\Delta$ P Wall Capillary | $N/m^2$ |
| PCM | $\Delta$ P Monogroove | " |
| PHEAD | $\Delta$ P Head Dia.$\rho\ell$ DVH | " |
| PTILT | $\Delta$ P Tilt Tilt | " |
| XLMG | | -- |
| CMGR | Monogroove Radial LC | -- |
| CMG | CMG $\leftarrow$ CMGR CMGR $\leftarrow$ 0.0 | |
| CVL | Laminar Vapor Channel LC | $\frac{kg}{kj \cdot m \cdot s}^2$ |
| RV | $v/\rho v \cdot \lambda$ Dummy Variable | |

## EQNA

| | | |
|---|---|---|
| CTOT | Rt Side of EQNA | $\frac{kg}{\cdot m \cdot s}^2$ |
| PNET | Lft Side of EQNA | $N/m^2$ |
| Q | Heat Transport | KJ/s |

## EQNB

| | | |
|---|---|---|
| PCMN | Lft Side of EQNB | $N/m^2$ |
| PVL | Rt Side of EQNA | $N/m^2$ |

## REYNOLD"S NO.

| | | |
|---|---|---|
| REN 1 | $D/\lambda v \cdot A \cdot \lambda$ | |
| REN | Reynold's Number REN 1*Q | $\frac{DQ}{A\lambda}$ |
| CVT | Turbulent Loss Coefficient | $\frac{kg}{jk \cdot m \cdot s}^2$ |
| QW | Heat Transport | Watts |
| RCWM | Transport Capacity | W • m |
| QSLW | Heat Transport at Sonice Limit | Watts |
| TCSLW | Transport Cap at Sonic Limit | W • m |
| QELW | Heat Transport @ Entrainment Limit | Watts |
| TCELW | Transport Cap @ Entrainment Limit | W • m |

FIGURE 2 FLOW CHART




COMPUTE BOTH SIDES
OF EQN B
PCMN = PCM - PTILT
PVL = (CA+CVL) * Q
Q MONOGROOVE
LIMITED
CHECK
EQN B
PCMN ≥ PVL?
NO
RECALCULATE Q
Q = PCMN / (CA+CVL)
REDEFINE PNET
PNET = PCMN
YES
WALL WICK LIMITED
2000
CALCULATE REYNOLD'S
NUMBER
REN = DVH*Q / μ·υ·λ·A υ
2500
E
IS
FLOW LAMINAR
REN < 2300?
YES
4050
C
IS
FLOW TURBULENT
REN > 3900
YES
3900
TURBULENT FLOW
CALCULATION
FTURB = .791 / Re^.25
FTRAN = FTURB
FLOW IS TRANSITIONAL
2300 ≤ Re 3900
FTRAN = (3.31 X 10^-5)Re^.69
CALCULATE TURBULENT
VAPOR L.C. CVT
CVT = Δ Pvapor channel turbulent / Q
EQN 6
4000

ORIGINAL PAGE IS
OF POOR QUALITY




IS
REN > 2300
YES
NO
PVL = (CA + CVL) *Q
IS
PCMN ≥ PVL
EQN B
YES
NO
Q MONOGROOVE
LIMITED
Q = PCMN/(CA+CVL)
PNET = PCMN
4060
CALCULATE QW +
TCWM, HEAT TRANS-
PORT + TRANSPORT
CAPACITY
CHECK SONIC LIMIT,
QSLW + SLTC
CHECK ENTRAINMENT
LIMIT, QELW + TCELW
PRINT OUT LOSS COEFFICIENTS
AND HEAT TRANSPORT RESULTS
END

```
            MATHEMATICAL MODELLING OF HEAT TRANSFER IN   DUAL PASSAGE HEAT PIPES

*************** OUTPUT DATA ******************
                  ***** FLUID PROPERTIES *****
OPERATING TEMP =      40.0000 C         STORAGE TEMP =           40.0000 C
CRITICAL TEMP =      132.4000 C         CRITICAL PRESSURE  =    112.9000 C
LATENT HEAT =       1101.0000 KJ/KG     THERMAL COND LIQ =        0.2720 W/M*C
LIQ DENSITY =        579.5000 KG/M^3    VAP. DENSITY =           12.0000 KG/M^3
LIQ. VISCOSITY =       0.2000 CPOISE    VAP. VISCOSITY =          0.0016 CPOISE
VAP SPEC HEAT =        2.1600 KJ/KG*C   LIQ SURF. TENSION =       0.0183 N/M
SAT. PRESSURE =       15.3400 BAR       SPEC. GAS CONST =         0.4882 KJ/KG*K

                  ***** HEAT PIPE GEOMETRY *****

EVAP LENGTH =      0.1500 M   COND LENGTH =      0.1500 M   TRANSFER LEN. =     0.4600 M
VAP CHAN HYD DIA =    0.013390 M   LIQ CHAN HYD DIA =   0.006320 M   O.D. =   0.023430 M
AREA VAP CHAN =   0.140816E-03 (M^2)    AREA LIQ CHAN =   0.313707E-04 (M^2)

WALL WICK OPENING: EVAP =   0.000051 M,  7559.0400TPM   COND =   0.000051 M,  7559.0400 TPM
MONOGROOVE OPENING =   0.000254 M,    LENGTH =   0.001240 M

    :

                  ***** LOSS COEFFICIENTS (KG/KJ*M*S^2) *****

MONOGROOVE RADIAL LOSS COEFF =  0.3173E+00
LIQ CHANNEL =   0.488327E+01WALL WICK EVAP =   0.187006E+04WALL WICK COND =   0.187006E+04VAPOR CHANNEL =   0.237458E+04

                  ***** CAPILLARY PRESSURES (KG/M*S^2) ****

WALL GROOVES     MONOGROOVE     TOTAL GRAVITY HEAD     NET CAP RISE     NET MONOGROOVE
  715.6428       144.3307           94.1703              621.4725          126.2812

REYNOLDS NUMBER =  8172.9731
WALL WICK STATIC WICKING HEIGHT =    112.4949 MM
MONOGROOVE STATIC WICKING HEIGHT =    25.3885 MM

TRANSPORT CAPACITY =       92.36 WATT*M    QMAX =     151.41 WATTS
 ***** WALL WICK LIMITED *****
```

FIGURE 3 PROGRAM OUTPUT

ORIGINAL PAGE IS
OF POOR QUALITY

## REFERENCES


1. Alario, J., Haslet, R. and Kosson, R., The Monogroove High Performance Heat Pipe, Proceedings of the AIAA 16th Thermophysics Conference, AIAA-81-1156, June 1981.

2. Chi, S.W., Mathematical Modeling of Cryogenic Heat Pipes, NASA CR-116175, September 1970; also see Paper No. 70-HT/SpT-6, ASME-AIChE Heat Transfer Conference, Minneapolis, Minn.,August 1969.

3. Chi, S.W., Mathematical Modeling of High and Low Temperature Heat Pipes, GW University Final Rept. to NASA, Grant No. NGR 09-101-070, December 1971.

4. Trush, R.B., Barker, R.S., O'Connor, E.W., Ayotte, W.J., Computer Simulation of the Life Support System for the Space Station Prototype, Proceedings of the ASME/SAE Life Support and Environmental Control Conference ASME 71-Av-34, July 1971.

5. Anand, D.K., On the Performance of a Heat Pipe, Engineering Notes, May 1966, ppg. 763.

6. Frank, S., Optimization of a Grooved Heat Pipe, Martin Marietta Corp., Baltimore, Md., 1971.

7. Holm, F.W., Miller, P.L., Thermal Scale Modeling of a Heat Pipe, Proceedings of ASME Space Technology and Heat Transfer Conference, Los Angeles, Calif., June 21-24, 1970.

ORIGINAL PAGE IS
OF POOR QUALITY

```
C -------------------------------------------------------------C
C                    DUAL PASSAGE HEAT PIPE ANALYSIS
C
C          MM      11/18/82
C
C ------------------------------------------------------------------------
      DIMENSION PROP(12)
      REAL LEFFM,LVM,NC,NE,OTEMPC
      PI = 3.1415926
      OPEN (UNIT=1,NAME='HEAT.OUT',TYPE='UNKNOWN')
      WRITE (1,10)
   10 FORMAT (' ',/,20X, 'MATHEMATICAL MODELLING OF HEAT TRANSFER IN
     + DUAL PASSAGE HEAT PIPES')
      WRITE(1,20)
   20 FORMAT(' ',25X,'*************** OUTPUT DATA *****************',/)
C
      ID = 1
      CALL FLUID(PROP,OTEMPC,ID)
      XLAT=PROP(1)
      RHOL=PROP(2)
      RHOV=PROP(3)
      TCONL=PROP(4)
      XMUL=PROP(5)
      XMUV=PROP(6)
      PSAT=PROP(7)
      CP  =PROP(8)
      STEN=PROP(9)
      TCRIT=PROP(10)
      PCRIT=PROP(11)
      RMW= PROP(12)
      DATA STEMPC, G / 40.0, 9.81 /
C
C ----------------- PRINT OUT FLUID PROPERTIES --------------------
      WRITE(1,100)
  100 FORMAT(' ',20X,'***** FLUID PROPERTIES *****')
      WRITE(1,110) OTEMPC,STEMPC
  110 FORMAT(' ','OPERATING TEMP = ',T20,F10.4,' C ',T40,
     + 'STORAGE TEMP = ',T60,F10.4,' C ')
      WRITE (1,120) TCRIT,PCRIT
  120 FORMAT(' ','CRITICAL TEMP = ',T20,F10.4,' C ',T40,
     + 'CRITICAL PRESSURE = ',T60,F10.4,' C ')
      WRITE(1,130) XLAT,TCONL
  130 FORMAT(' ','LATENT HEAT = ',T20,F10.4,' KJ/KG',T40,
     + 'THERMAL COND LIQ = ',T60,F10.4,' W/M*C')
      WRITE(1,140) RHOL,RHOV
  140 FORMAT(' ','LIQ DENSITY = ',T20,F10.4,' KG/M^3 ',T40,
     + 'VAP. DENSITY = ',T60,F10.4,' KG/M^3')
      WRITE(1,150) XMUL,XMUV
  150 FORMAT(' ','LIQ. VISCOSITY = ',T20,F10.4,' CPOISE ',T40,
     + 'VAP. VISCOSITY = ',T60,F10.4,' CPOISE')
      WRITE(1,160) CP,STEN
  160 FORMAT(' ','VAP SPEC HEAT = ',T20,F10.4,' KJ/KG*C ',T40,
     + 'LIQ SURF. TENSION = ',T60,F10.4,' N/M ')
      WRITE(1,170) PSAT,RMW
  170 FORMAT(' ','SAT. PRESSURE = ',T20,F10.4,' BAR ',T40,
     + 'SPEC. GAS CONST = ',T60,F10.4,' KJ/KG*K ')
C
C ------------------------------------------------------------------------
C              INPUT GEOMETRY
```

```
      DATA EVAPM,TRANSM,CONDM,DV,DL,TW /0.15,0.46,0.15,13.39,6.32,1.24/
C   YIELD STRENGTH, LIQUID AREA FRACTION, WETTING ANGLE
      DATA YS,AZ,THETWR,TILT /137.9E+06,1,0.0,3.175/
C
C   INPUT WALL WICK PROPERTIES
C   WEB WIDTH, ROOT WIDTH, GROOVE DEPTH, THREADS/INCH, NO. OF FEEDS
      DATA WWE,RWE,GDE,TPIE,NE /0.051,0.014,0.196,192.0,1.0/
      DATA WWC,RWC,GDC,TPIC,NC /0.051,0.014,0.196,192.0,1.0/
C   FORM FACTOR FOR HEAT INPUT
      DATA QZ /1.0/
C
C   INPUT MONOGROOVE PROPERTIES
C   GROOVE WIDTH, TAPER, STAND OFF DISTANCE
      DATA WM,ALPHAD,DELT /0.254,0.0,1.24/
C
C               CONVERT INPUT VARIABLES FROM (MM) TO (M)
C
      DV = DV/1000.
      DL = DL/1000.
      TW = TW/1000.
      WWE = WWE/1000.
      RWE = RWE/1000.
      GDE = GDE/1000.
      WWC = WWC/1000.
      RWC = RWC/1000.
      GDC = GDC/1000.
      WM = WM/1000.
      DELT = DELT/1000.
      TILT = TILT/1000.
C
C   CALCULATE GEOMETRY
C
C   1. PIPE GEOMETRY
      LEFFM = TRANSM+(EVAPM+CONDM)/2
      LVM = TRANSM + EVAPM + CONDM
C
C   2. WALL GROOVE GEOMETRY (SIMPLIFIED)
C      FOR A FIRST APPROX. ASSUME GROOVE FILLED WITH LIQUID
C      (IE. AWSEG=0.)
      IXX = 0
      WW = WWE
      RW = RWE
      GD = GDE
   40 XX = (WW-RW)/(2*GD)
      ALPHTR = ATAN(XX)
      AW = GD*(WW+RW)/2.0
      WPW = RW+(2*GD)/COS(ALPHTR)
      DW = (4*AW)/WPW
      IF (IXX .NE. 0) GO TO 50
      AWE = AW
      WPWE = WPW
      DWE = DW
      WW = WWC
      RW = RWC
      GD = GDC
      IXX = 1
      GO TO 40
   50 AWC = AW
      WPWC = WPW
      DWC = DW
C
C   3. MONOGROOVE GEOMETRY
      OD = DV + DL + (2*TW) + DELT
C
C   4. HYDRAULIC DIAMETERS
C      ASSUME AZ = 1 (100% FULL)
```

```
C     A. LIQUID HYD DIA
      WPL = PI*DL*AZ
      AL = WPL*DL/4
      DLH = DL
C
C     B. VAPOR HYD. DIA.
C        UNTIL DEFINITION OF NE IS KNOWN
      WPV = PI*DV
      AV = WPV*DV/4
      DVH = DV
C
C  CHECK IF DVH > DLH
      IF(DVH.GT.DLH) GO TO 70
      WRITE(1,60)
      TYPE 60
   60 FORMAT(' ',/,'***** ERROR-DIAMETER OF LIQUID CHANNEL EXCEEDS
     + DIAMETER OF VAPOR CHANNEL *****',/)
   70 CONTINUE
C
C       CHECK PRIMING CAPABILITY OF VOLUME
C
      AL1 = (PI*DL*DL/4.0)+(DELT*WM)
      DL1 = SQRT(4*AL1/PI)
      AV1 = PI*DV*DV/4.0
      AV2 = AV1-AL1
      DV2 = SQRT(4.0*AV2/PI)
      IF (DL1 .GT. DV2) GO TO 200
      WRITE (1,80)
   80 FORMAT(' ',20X,'***** WILL NOT PRIME IN ZERO-G *****')
      GO TO 6000
C     RETURN
  200 CONTINUE
C
      TPME = TPIE*39.37
      TPMC = TPIC * 39.37
C  CONVERT TPI TO THDS/M (TPM)
C
C --------------------- PRINT OUT GEOMETRY ------------------------------
C    OUTPUT HEAT PIPE GEOMETRY
      WRITE(1,300)
  300 FORMAT(' ',//,20X,'***** HEAT PIPE GEOMETRY *****',/)
      WRITE(1,310) EVAPM,CONDM,TRANSM
  310 FORMAT(' ','EVAP LENGTH = ',F10.4,' M   COND LENGTH = ',F10.4,
     +' M   TRANSFER LEN. = ',F10.4,' M ')
      WRITE(1,320) DVH,DLH,OD
  320 FORMAT(' ','VAP CHAN HYD DIA = ',F10.6,' M   LIQ CHAN HYD DIA = ',
     + F10.6,' M   O.D. = ',F10.6,' M ')
      WRITE(1,325) AV,AL
  325 FORMAT(' ','AREA VAP CHAN = ',E14.6,' (M^2)   AREA LIQ CHAN = ',
     + E14.6,' (M^2) ',/)
      WRITE(1,330) WWE,TPME,WWC,TPMC
  330 FORMAT(' ','WALL WICK OPENING: EVAP = ',F10.6,' M, ',F10.4,
     + 'TPM   COND = ',F10.6,' M, ',F10.4,' TPM ')
      WRITE(1,340) WM,DELT
  340 FORMAT(' ','MONOGROOVE OPENING = ',F10.6,' M,   LENGTH = ',F10.6,
     + ' M ',/)
C
C ------------------- CALCULATE LOSS COEFFICIENTS ----------------------
C
C   1. CA - LIQUID CHANNEL LOSS COEFF. (LC)
C
      RL = XMUL/(RHOL*XLAT*1000.)
      CA = (32*RL*LEFFM)/(AL*DLH*DLH)
C
C   2. CW - WALL WICK LC
      CW = 32*PI*RL*DVH*QZ
```

ORIGINAL PAGE IS OF POOR QUALITY

```
      CWE = CW/(8*NE*EVAPM*TPME*AWE*DWE*DWE)
      CWC = CW/(8*NC*CONDM*TPME*AWC*DWC*DWC)
      CW = CWE + CWC
C
C   3. PCW - DELTA P WALL CAPILLARY
      PCW = 2*STEN*COS(THETWR+ALPHTR)/WWE
C
C   4. PCM - DELTA P MONOGROOVE
      PCM = 2*STEN*COS(THETWR)/WM
C
C   5. PHEAD - DELTA P HEAD DIA. IN GRAVITY
      PHEAD = RHOL*DVH*G
C
C   6. PTILT - DELTA P TILT IN GRAVITY
      PTILT = RHOL*TILT*G
C   CHECK FOR 0-G. (TILT < -100)
      IF (TILT .GT. -100.) GO TO 1000
      PHEAD = 0.0000
      PTILT = 0.0000
 1000  CONTINUE
      PHD = PHEAD + PTILT
C ------------------------------------------------------------------------
C
C   EQNS 1,2,3,4,7 &,8 HAVE BEEN COMPUTED
C   NOW IT MUST BE DETERMINED IF FLOW IS LAMINAR
C   (FOR EQN 5) OR TURBULENT (FOR EQN 6)
C   FIRST ASSUME LAMINAR FLOW, CALCULATE Q
C   THEN CALC. REN AND CHECK.
C
C   8. CVL - LAMINAR VAPOR CHANNEL LOSS COEFFICIENT
C
      RV = XMUV/(RHOV*XLAT*1000.0)
      CVL = (32*RV*LEFFM)/(AV*DVH*DVH)
C
C ----------------- HEAT TRANSPORT CALCULATIONS ---------------------
C
C
C   COMPUTE Q FROM EQN A
      CTOT = CVL+CA+CW
      PNET = PCW-PHD
      Q = PNET/CTOT
C   NOW CHECK EQN B
      PCMN = PCM-PTILT
      PVL = (CA+CVL)*Q
      IF (PCMN .GE. PVL) GO TO 2000
      Q = PCMN/(CA+CVL)
      PNET = PCMN
C
C   CALCULATE REYNOLD'S NUMBER
C
 2000 REN1 = (DVH*1000.0)/(XMUV*AV*XLAT)
 2300 REN = REN1*Q
      TYPE *, REN
C
C   CHECK REYNOLD'S NO. FOR TURBULENT OR TRANSITION FLOW
 2500 IF (REN .LT. 2300.) GO TO 4050
      IF (REN .GT. 3900.) GO TO 3900
C
C   TRANSITION FLOW CALCULATIONS
      FTRAN = (3.31E-5)*(REN**.69)
      GO TO 4000
C
C   TURBULENT FLOW CALCULATIONS
 3900 FTURB = .0791/(REN**.2500)
      FTRAN = FTURB
C
```

```
C   TURBULENT VAPOR LOSS COEFFICIENT
 4000 CVT = (2*FTRAN*LEFFM)/(AV*AV*DVH*RHOV*XLAT*XLAT)
C
C   COMPUTE Q FROM EQN A FOR TURBULENT FLOW
      IF (PNET .GT. PCMN) GO TO 4030
      B = CA/CVT
      GO TO 4040
 4030 B=(CA+CW)/CVT
 4040 XX = (B*B)+4*(PNET/CVT)
      XX = SQRT(XX)
      Q1 = (XX-B)/2.0
C   IF DIFFERENCE BETWEEN NEW Q (CAUSED BY TURB. FLOW
C   CALCULATIONS) AND PREVIOUSLY CALCULATED Q IS >1%
C   THEN RECALC. REYNOLD'S NO. USING NEW Q AND REPEAT
C   UNTIL DIFF < 1%
C
      DIFF = ABS((Q1-Q)/Q)
      Q = Q1
      IF (DIFF.GE. .010) GO TO 2300
C
C   CHECK EQN B
      PVLT = Q*(CA+(CVT*Q))
      IF (PCMN .GE. PVLT) GO TO 4050
      ZY = (CA*CA) + 4.0*CVT*PCMN
      ZY = SQRT(ZY)
      Q = (ZY-CA)/(2.0*CVT)
      PVL = PVLT
      PNET = PCMN
      GO TO 2300
 4050 CONTINUE
C
C   REPEAT LAMINAR EQUATIONS
      REN = REN1*Q
      IF (REN .GT. 2300.) GO TO 4060
      PVL = (CA+CVL)*Q
      IF (PCMN .GE. PVL) GO TO 4060
      Q = PCMN/(CA+CVL)
      PNET = PCMN
 4060 CONTINUE
      QW = Q*1000.0
      TCWM = QW*LEFFM
C
C   END HEAT TRANSPORT CALCULATIONS
C
C
C                  CHECK SONIC LIMIT
C
      GAMMA = 1.33
      TEMPK = OTEMPC + 273
      VS = SQRT(1000.0*GAMMA*RMW*TEMPK/(2.0*GAMMA+2.0))
      QSLW = RHOV*XLAT*AV*VS*1000.0
      SLTC = QSL * LEFFM
C
C                   CHECK ENTRAINMENT LIMIT
C
      XX = (RHOV*STEN*XLAT*XLAT)/(WM/(2*COS(THETWR)))
      QELW = AV*SQRT(XX)*1000.0
      TCELW=QELW*LEFFM
C
C -------------- PRESSURE AND LOSS COEFFICIENT OUTPUT -----------------
      WRITE (1,400)
  400 FORMAT(' ',//,20X,'***** LOSS COEFFICIENTS (KG/KJ*M*S^2) *****',/)
      WRITE (1,410)
  410 FORMAT(' ','LIQUID CHANNEL       VAPOR CHANNEL       WALL WICK
     + WALL WICK EVAP      WALL WICK COND')
      IF (REN.GE.2300.) GO TO 5000
```

ORIGINAL PAGE IS
OF POOR QUALITY

```
      WRITE (1,420) CA,CVL,CW,CWE,CWC
      GO TO 5100
 5000 WRITE (1,420) CA,CVT,CW,CWE,CWC
 5100 CONTINUE
  420 FORMAT(' ',E14.6,4X,E14.6,2X,E14.6,2X,E14.6,4X,E14.6)
      WRITE (1,430)
  430 FORMAT(' ',/,20X,'***** CAPILLARY PRESSURES (KG/M*S^2) ****',/)
      WRITE (1,440)
  440 FORMAT (' ','WALL GROOVES      MONOGROOVE      TOTAL GRAVITY HEAD
     +     NET CAP RISE      NET MONOGROOVE')
      WRITE (1,450) PCW,PCM,PHD,PNET,PCMN
  450 FORMAT(' ',F10.4,5X,F10.4,8X,F10.4,12X,F10.4,9X,F10.4,//)
      WRITE (1,460) REN
  460 FORMAT (' ','REYNOLDS NUMBER = ',F10.4)
C -----------------------------------------------------------------------
C
C                          PRINT TRANSPORT CAPACITY
C
      WRITE (1,490) TCWM,QW
  490 FORMAT (' ','TRANSPORT CAPACITY = ',F10.2,' WATT*M   QMAX = ',
     + F10.2,' WATTS ')
      IF (PCMN.GT.PVL) GO TO 5200
      WRITE (1,500)
  500 FORMAT('+','  ***** MONOGROOVE LIMITED *****')
      TYPE *,'***** MONOGROOVE LIMITED *****'
      GO TO 5300
 5200 WRITE (1,510)
  510 FORMAT('+','  ***** WALL WICK LIMITED ***** ')
      TYPE *,'***** WALL WICK LIMITED *****'
 5300 IF(QW .LE. QSLW) GO TO 5400
      WRITE (1,520)
  520 FORMAT('+','  ***** SONIC LIMIT EXCEEDED *****  ')
      WRITE (1,530) QSLW,SLTC
  530 FORMAT(' ',20X,'LIMITS ARE: Q = ',T30,F10.2,' WATTS',T60,
     + 'TRANSPORT CAPACITY = ',T80,F10.2,' WATT-M')
 5400 IF (QW .LE. QELW) GO TO 6000
      WRITE (1,540)
  540 FORMAT('+',' ***** ENTRAINMENT LIMIT EXCEEDED *****  ')
      WRITE (1,530) QELW,TCELW
 6000 CONTINUE
      CLOSE (UNIT=1)
      STOP
      END
C *******************************************************************
C     7
C                        FLUID PROPERTIES SUBROUTINE
C
C -------------------------------------------------------------------
C                        DATA FORMAT
C
C *******************************************************************
C          *DAT(1-10,1,X) - TCRIT,PCRIT,TMP,TBP,RMW,TDMIN,TDMAX,0,0,0,
C          *    TEMP -60,-40,-20,0,20,40,60,80,100,120
C          *PROPERTY      DAT(X,#,X)     PROP(#)      UNITS
C          *    XLAT           2             1        KJ/KG
C          *    RHOL           3             2        KG/M3
C          *    RHOV           4             3        KG/M3
C          *    TCONL          5             4        KJ/M DEG.C
C          *    XMUL           6             5        CENTIPOISE
C          *    XMUV           7             6        CENTIPOISE
C          *    PSAT           8             7        BAR
C          *    CP             9             8        KJ/KG DEG.C
C          *    STEN          10             9        N/M
C          *   TCRIT          1,1           10        DEG. C
C          *   PCRIT          2,1           11        BAR
C          *     RMW          3,1           12        KJ/KG*DEG.
```

ORIGINAL PAGE IS
OF POOR QUALITY

```
C         *********
C         *
C         *
C         *
C FLUID#? *  ETC.
C
C ****************** *****************************************
      BLOCK DATA
      DIMENSION AMMON(10,10),DAT(10,10,1)
      DIMENSION PROPS1(10),XLAT1(10),RHOL1(10),RHOV1(10),TCON1(10)
      DIMENSION XMUL1(10),XMUV1(10),PSAT1(10),CP1(10),STEN1(10)
      COMMON /AMMON/ PROPS1, XLAT1, RHOL1, RHOV1,
     +   TCON1, XMUL1, XMUV1, PSAT1, CP1, STEN1
      EQUIVALENCE (AMMON(1,1),PROPS1)
      EQUIVALENCE (DAT(1,1,1),AMMON)
C          ENTER DATA IN THE FOLLOWING BLOCK ACCORDING TO ABOVE FORMAT
C
C2345 /------------- STATEMENT FIELD (72 SPACES) ----------------------\
C
C                           AMMONIA PROPERTIES
C
       DATA PROPS1 /132.4,112.9,-77.7,-33.4,.4882,-60,120,0,0,0/,
     2 XLAT1 /1434.,1384.,1338.,1263.,1187.,1101.,1026.,891.,699.,428./,
     3 RHOL1 /714.4,690.4,665.5,638.6,610.3,579.5,545.2,505.7,455.1,
     4        374.4/,
     5 RHOV1 /.03,.05,1.62,3.48,6.69,12.0,20.49,34.13,54.92,113.16/,
     6 TCON1 /.294,.303,.304,.298,.286,.272,.255,.235, .212, .184/
       DATA XMUL1 /.360,.290,.260,.250,.220,.200,.170,.150,.110,.70/,
     2 XMUV1 /.0072,.0079,.0085,.0092,.0101,.0016,.0127,.014,.016,
     3 .0189/,
     4 PSAT1 /.27,.76,1.93,4.24,8.46,15.34,29.80, 40.9, 63.12, 90.44/,
     5 CP1 /2.05,2.075,2.1,2.125,2.15, 2.16, 2.18, 2.21, 2.26, 2.92/
       DATA STEN1 /.04062,.03574,.0309,.0248,.02133,.01833,.01367,
     1 .00767,.005,.0015 /
C
C                            ACETONE PROPERTIES
C
      END
C ---------------------------------------------------------------
C
C                     FLUID PROPERTY CALCULATION SUBROUTINE
C
C
      SUBROUTINE FLUID(PROP,OTEMPC,ID)
      DIMENSION DAT(10,10,1), PROP(9), AMMON(10,10)
      COMMON /AMMON/ X1(10),X2(10),X3(10),X4(10),X5(10),
     +               X6(10),X7(10),X8(10),X9(10),X10(10)
      EQUIVALENCE (DAT,AMMON)
      EQUIVALENCE (AMMON,X1)
C
  100 TYPE *,'INPUT OPERATING TEMP IN DEG. C. '
  120 ACCEPT *, OTEMPC
      TYPE *,'OTEMPC= ',OTEMPC
      TYPE  10, ((DAT(I,J,1),I=1,10),J=1,10)
   10 FORMAT(' ',10F8.3)
C
C          CHECK TEMP TO SEE OTEMPC > TCRIT
      IF (OTEMPC .LT. DAT(1,1,ID)) GO TO 210
      TYPE *,'**** WORKING TEMP EXCEEDS TCRIT **** '
      GO TO 110
C
C          CHECK FOR TEMP > FREEZING
  210 IF (OTEMPC .GT. DAT(3,1,ID)) GO TO 250
      TYPE *,'**** TEMPERATURE BELOW FREEZING POINT ****'
      GO TO 110
C
```

```
C           CHECK FOR OTEMPC IN RANGE OF DATA - TDMIN < OTEMPC < TDMAX
  250 IF (OTEMPC .LT. DAT(6,1,ID)) GO TO 300
      IF (OTEMPC .GT. DAT(7,1,ID)) GO TO 300
      GO TO 350
  300 TYPE *,'**** INSUFFICIENT DATA - TEMP OUT OF DATA RANGE ****'
      TYPE *, DAT(6,1,ID),' - ',DAT(7,1,ID)
  310 GO TO 110
C
C           CALCULATE INDEX NO.
  350 XNO = (OTEMPC/20.) + 4.
      NM1 = INT(XNO)
        N = NM1 + 1
        R = N - XNO
      DO 400 I = 1,9
          II = I + 1
          PROP(I) = DAT(N,II,ID)-(R*(DAT(N,II,ID)-DAT(NM1,II,ID)))
  400 CONTINUE
      PROP(10) = DAT(1,1,ID)
      PROP(11) = DAT(2,1,ID)
      PROP(12) = DAT(5,1,ID)
      GO TO 450
  110 TYPE *,'INPUT NEW TEMP (DEG. C) '
      GO TO 120
  450 CONTINUE
      RETURN
      END
```

ORIGINAL PAGE IS
OF POOR QUALITY